



Creating ypinator™ Scripts

1.0

for Typinator 5.0 or newer, January 2012

**Typinator™ is a macility® product
© 2012 ergonis software**

Contents

- ▶ About this document2
- ▶ Integration of scripts in Typinator2
- ▶ AppleScript files3
- ▶ Simple scripts3
- ▶ Scripts with parameters.....3
- ▶ Testing scripts4
- ▶ Other scripting languages5
- ▶ Compiled command-line tools6
- ▶ Important rules6
- ▶ Environment variables7
- ▶ Returning values to Typinator9
- ▶ Calculator variables9
- ▶ Simulating backspaces10
- ▶ Canceling expansions11
- ▶ Contributions are welcome11

About this document

Starting with version 5.0, Typinator has the ability to execute external scripts and include the result of the script execution in expansions. Using such scripts is fairly easy. When you edit an expansion, just open the {...} pop-up menu next to the expansion field and select the desired script. Typinator inserts a special marker that contains a script invocation, which will be executed when the expansion is about to be inserted in a document.

This document is meant for those who wish to create new scripts. It describes the mechanism and explains how scripts should be written to use the full potential of Typinator's scripting feature. It assumes that you are already familiar with programming in AppleScript, shell scripting, Perl, PHP, Python, Ruby, or with the development of command line tools in C or Objective C.

This document is accompanied with a few samples that you can install and modify. Some of these samples are mentioned in the text, so we recommend that you put the "Samples" and "Hello" folders into the "Includes" folder.

IMPORTANT: We renamed two sample scripts since the first version of this document:

The Convert script is now called Multiply.

The Multiple script is now called Repeat.

Please replace the Samples folder with the new attached version. If you already used the Convert and Multiple scripts in expansions, please replace invocations with the new names.

Integration of scripts in Typinator

Typinator uses special placeholders of the form {xy} in expansions. These placeholders get replaced with dynamic text which is determined at the time when the expansion takes place. For example, {h24}:{m} gets replaced with the hour and minute of the current time. Typinator uses the same notation and technique for scripts. A script invocation uses the notation

{folder/scriptName argument}

where the argument is optional. When the first part (folder/scriptName) refers to an executable script inside the Includes folder, Typinator invokes the script with the given parameter and replaces the entire placeholder with the result of the script invocation.

Scripts are stored in the Includes folder, which in turn resides in Typinator's sets folder. To quickly open the Includes folder, use the command at the bottom of the {...} menu. We recommend to store scripts not at the top level of the Includes folder, but rather inside a nested folder. Nested folders appear as submenus inside the {...} menu and make it easier to organize scripts by topics.

AppleScript files

To create an AppleScript file for use with Typinator, we recommend that you use AppleScript Editor and save the script as a text file in a subfolder within the Includes folder. Make sure to save the script with the extension `.applescript`. Typinator does not recognize compiled scripts, as it needs to look at the script's source text.

Typinator strips off the extension in the `{...}` menu, but it includes the extension in the generated placeholder. For example, the script file `IP.applescript`, stored in the `Scripts` subfolder, appears as `IP` in the menu and is inserted as `{Scripts/IP.applescript}`.

Simple scripts

When Typinator uses an expansion with a script placeholder, it runs the script and replaces the placeholder with the result of the script. The script therefore needs to return the desired result. If the result is not text, Typinator tries to coerce it to text.

A very simple script could look like this:

```
return "Sample Text"
```

A more meaningful script for inserting the computer's IP number is:

```
return IPv4 address of (system info)
```

Scripts with parameters

Typinator can invoke scripts with parameters. Parameters are simple strings that follow the script name in the placeholder. To create a script that accepts parameters, it must contain a handler with the name `expand` and a single parameter. For example, the following script takes a string parameter and returns the length of the string:

```
on expand(str)
    return length of str
end expand
```

If the script has the name `Length`, you can invoke it as

```
{Scripts/Length.applescript parameter}
```

You can even nest script invocations. For example,

```
{Scripts/Length.applescript {Scripts/IP.applescript}}
```

returns the number of characters of the current IP address.

When you create a string with a parameter, we recommend that you include a hint about the meaning of the parameter in the script. You can do this by means of a comment on the same line as the header of the `expand` handler. The comment must begin with `parameter:`:

```
on expand(str) -- parameter: any string
    return length of str
end expand
```

When you select the *Length* script from the menu, Typinator includes the hint as a default parameter in the placeholder:

```
{Scripts/Length.applescript any string}
```

When a script expects a keyword parameter that controls the behavior of the script, list the possible alternatives, separated by vertical bars, as shown in the *UserName* example:

```
on expand(kind) -- parameter: short|long|computer
    if kind = "short" then
        return (short user name of (system info))
    else if kind = "computer" then
        return (computer name of (system info))
    else -- "long" or anything else
        return (long user name of (system info))
    end if
end expand
```

The corresponding placeholder now looks like this:

```
{Samples/UserName.applescript short|long|computer}
```

This makes it pretty clear what the script expects. Using arguments in this way makes it easy to create a single script for multiple related purposes.

Note that Typinator supports only a single parameter. If your script needs multiple parameters, you need to split the parameter into parts yourself. Here is a sample code fragment that splits a single parameter `theParameter` into an array of parameters, using a space character as a separator.

```
set originalDelimiters to AppleScript's text item delimiters
set AppleScript's text item delimiters to " "
set parameterList to every text item of theParameter
set AppleScript's text item delimiters to originalDelimiters
```

When you use such a technique, also supply a default parameter that makes clear what the script expects.

Testing scripts

You can test your scripts independently of Typinator in AppleScript Editor. Simply run your script and check the result in the AppleScript Editor window. If your script has an *expand* handler, just include an invocation of the handler in the script. For example, you could test the *UserName* script as follows:

```
return expand("long") -- for testing in the AppleScript Editor

on expand(kind) -- parameter: short|long|computer
    if kind = "short" then
        return (short user name of (system info))
    else if kind = "computer" then
        return (computer name of (system info))
    else -- "long" or anything else
        return (long user name of (system info))
    end if
end expand
```

Other scripting languages

AppleScript is only a special case; Typinator can actually run any script that you could invoke from the command line in Terminal. This includes Bash, Perl, PHP, Python, and Ruby.

Files containing scripts developed in these languages do not need to end with a special extension, but they must have the *executable* attribute set. To set this attribute, use the Unix command

```
chmod +x filename
```

in the Terminal. You will typically also use a Unix shell like the Terminal application to test your scripts.

To specify the scripting language, the script must begin with a “shebang” line that contains the path to the script interpreter. For example, to run a Python script, the first line should look like this:

```
#!/usr/bin/python
```

As in AppleScript, you can include a comment that contains “-- parameter:” to give Typinator a hint about expected arguments. In most scripting languages, comments start with a hash (#) character, so a typical parameter comment would look like this:

```
#-- parameter: name
```

Note that a single space character must be present between “--” and “parameter”.

Your scripts can do almost anything; to define the resulting expansion, just write the desired result to standard output. Typinator will interpret the output as UTF-8.

The “Hello” that comes with this manual contains a few scripts in different scripting languages. All these scripts do the same thing: They just display the message “Hello, World!”. If invoked with a parameter, they display the parameter in place of the word “World”. Take a look at these examples to see how to process parameters and return a result from scripts. Here is the contents of the Bash sample:

```
#!/bin/bash
#-- parameter: name
if [ "$1" != "" ]; then
    echo -n "Hello, $1!" # -n suppresses newline
else
    echo -n "Hello, World!"
fi
```

To include this script in an expansion, you would, for example, write

```
HAL said "{Hello/HelloBash Dave}".
```

to produce the following expansion:

```
HAL said "Hello, Dave!".
```

Compiled command-line tools

As yet another possibility, you can create precompiled command-line tools, written in C or Objective C. The “HelloTool” example is a simple C tool with the following source text:

```
#include <stdio.h>
const char * parameter = "\n-- parameter: name\n";
int main (int argc, const char * argv[]) {
    if (argc > 1) {
        printf("Hello, %s!", argv[1]);
    } else {
        printf("Hello, World!");
    }
    return 0;
}
```

To create an executable command-line tool, you would typically use Xcode and create a “Command Line Tool” project.

Since the resulting tool contains only executable code, you must use a trick to specify the expected parameter: Simply declare a string constant that contains “-- parameter:” and ends with a newline character. The compiler will then include the parameter description in the executable file, and Typinator will be able to extract it.

Important rules

When writing scripts, you should consider that Typinator invokes them in a low-level situation while the user is typing text. Scripts therefore should **execute quickly**, typically in less than a second. For command-line tools and scripting languages other than AppleScript, Typinator limits the available time to 5 seconds. If your script takes longer, Typinator terminates the script without a warning.

This also means that scripts **should not interact with the user**. Do not use any alerts or dialogs inside your scripts.

It is also a good idea to **check for errors**. If your script might fail, use an exception handler, if possible. In AppleScript, wrap the critical parts in a try clause and return an error message in the error handler.

Environment variables

Typinator sets a few environment variables to tell scripts about the current context. These are:

- *Typinator_version*: The version number of Typinator.
- *Typinator_setsFolder*: The path to the Sets folder, as specified in Typinator's preferences.
- *Typinator_abbreviation**: The abbreviation that triggered the expansion.
- *Typinator_wholeWord*: The "whole word" setting of the abbreviation (0 or 1).
- *Typinator_formatted*: 1 for formatted text expansion, 0 for plain text.
- *Typinator_language*: The language used for date elements in the expansion (such as "en" for English)
- *Typinator_set**: The name of the set that contains the abbreviation.
- *Typinator_matchingText**: The entered text that matches the abbreviation. This may differ from *Typinator_abbreviation* in case. For example, "Jgv" matches "jgv" when the abbreviation is defined with the option "Case does not matter".
- *Typinator_precedingText**: The text that was entered immediately before the abbreviation. Note that this is only the typed text that Typinator knows about. When the user clicks in the middle of a paragraph and then types an abbreviation, Typinator does not know anything about the text before the insertion point. *Typinator_precedingText* will then be the empty string.
- *Typinator_date*: The current date (considering the *yearDelta*, *monthDelta* and *dayDelta* settings) in the form year-month-date, as it would be inserted with the marker combination {YYYYY}-{MM}-{DD}.
- *Typinator_application**: The name of the current application in which the expansion will be inserted.
- *Typinator_applicationPath*: The full path of the application. (such as "/Applications/TextEdit.app")
- *Typinator_applicationIdentifier*: The unique identifier of the application. (such as "com.apple.TextEdit").
- *Typinator_applicationVersion*: The application's version number.

* Environment variables are stored in an 8-bit encoding (MacRoman for AppleScript, UTF-8 for shell scripts). Special characters in text (such as accented letters) may get lost during the translation into 8-bit format.

You can access these variables like any other environment variables. For example, prefix the name with a dollar sign in Bash:

```
#!/bin/bash
echo "Typinator version: $Typinator_version"
```

In AppleScript, use "system attribute" to access the variables:

```
set TypiVersion to system attribute "Typinator_version"
```

For examples in AppleScript and Bash that list Typinator's special environment variables, see the ContextAS and ContextBash scripts in the Samples folder.

Typinator uses the same mechanism also for the date and time offsets. For example, when you insert a marker for a date calculation that sets the date to tomorrow, Typinator uses the notation `{{dayDelta=+1}}`. This creates an environment variable "dayDelta" with the value "+1". Typinator uses these values internally for date markers, but also defines them as environment variables, so scripts can access these values. The following example shows how you can find out whether there is a day offset:

```
set dayDelta to system attribute "dayDelta"
if dayDelta = "" or dayDelta = "0" then
    return "today"
else if dayDelta = "+1" then
    return "tomorrow"
end if
```

Note that the variables are represented by strings. To use the numeric value, you need to coerce the value to type *integer*.

The notation with the double braces is actually more general than that. It allows you to define *any* variables. For example, you could include something like `{{magicNumber=42}}` or `{{one=1;two=2}}` or `{{server="www.ergonis.com"}}` in an expansion. These assignments are processed left-to-right and are preserved across expansions and even restarts.

You can use such assignments to implement states that you can modify with simple abbreviations. For example, you can create two abbreviations `offerOn` and `offerOff` that do not actually expand into anything but just turn a state variable on and off: `{{specialOffer=1}}` and `{{specialOffer=0}}`. You can then use these abbreviations to control the behavior of scripts that produce different results depending on the `specialOffer` variable.

Note that the `{{...}}` assignments should not contain any spaces. When you write `{{v=12}}` or `{{x=first;y=second;}}`, Typinator treats everything between the equals sign and the closing braces or semicolon as the string to be assigned. When you write `{{v= 12}}`, the string " 12" (digits 1 and 2 with a leading space character) is assigned to `v`.

Note: To avoid conflicts with common Unix environment variables (such as `HOME`, `PATH`, `SHELL`, `TMPDIR`, and `USER`), variables used in Typinator must contain at least one lower-case character. Typinator ignores all variables whose names consist only of capital letters and "_".

Returning values to Typinator

You can use the same notation to return variables to Typinator. Just include a `{{...}}` block *at the beginning* of the returned result and embed assignments between the braces. Typinator removes the `{{...}}` block and handles the assignments inside. As mentioned above, the variables' names must contain at least one lower-case letter.

Typinator stores the returned variables in a global dictionary and keeps them around in its preferences. This means that scripts can use this mechanism to permanently store values and use them during a subsequent invocation. For an example, see the *Counter* script:

```

on expand(counterName) -- parameter: counterName
  -- if no counter name is given, use "default_counter"
  if counterName = "" then set counterName to "default_counter"
  -- get the value of the counter
  set counterString to system attribute counterName
  set counterValue to 0
  try
    set counterValue to counterString as integer
  end try
  set counterValue to counterValue + 1
  -- set the result to a {{...}} block that returns the new counter value
  -- and append the counter value to return it as the result of the expansion
  return "{" & counterName & "=" & counterValue & "}" & counterValue
end expand

```

You can now embed this script in an abbreviation like this:

```
expanded {Samples/Counter.applescript myCounter} times
```

This creates a new variable "myCounter" and increments it on each invocation. You can use this mechanism with as many counters as you like, and you can easily reset a counter in another expansion using `{{myCounter=0}}`.

You can even include the counter value in an expansion without running any script by simply writing the variable name inside a pair of double braces: `{{myCounter}}`. This placeholder (note the absence of an equals sign) gets replaced with the variable's value.

You can use this mechanism to create date and time calculations. For example, you can use a script that returns `{{yearDelta=0;monthDelta=0;dayDelta=+7}}` to set the date one week ahead of the current date. For more elaborate calculations, see the Advanced Date Math package that is available on our "[Download Extras](#)" web page.

Calculator variables

The calculator in the Quick Search field has support for variables. For example, you can enter a calculation like

```
total=120.5+67.3+45
```

When you insert the result in a document, you get 232.8, and Typinator also assigns this value to a variable named *total*. To add 25%, you can later compute

```
total*1.25,
```

which results in the value 291.

These variables are available as environment variables with the prefix "calculator_". In AppleScript, the variable *total* can be accessed as

```
set total to system attribute "calculator_total"
```

To modify the value of a calculator variable, include an assignment in a `{{...}}` block, as described in the previous section. For example, the following statements define a new variable *amount* that is computed as *total* plus 25%:

```
set amount to total * 1.25
return "{{calculator_amount=" & amount & "}}"
```

After execution of this script, you can use the variable *amount* in the calculator.

You can also use the `{{...}}` notation to include a calculator variable in an expansion without running a script: `{{calculator_amount}}`.

Simulating backspaces

Typinator uses a special variable named *bs* to trigger a series of backspace keystrokes before inserting an expansion. In combination with the *Typinator_precedingText* environment variable, you can use this trick for a few cool effects. For example, you can create a script that looks for a number immediately to the left of the typed abbreviation. The script can use this number in its computation of the result, and it can return the number of desired backspaces in the *bs* variable to make Typinator "eat up" the number before inserting the result. For an example of this technique, see the *Repeat* script. This script takes a string as a parameter and repeats this string as often as specified by the number before the typed abbreviation. You can then create an abbreviation `*-` with the following expansion:

```
{Samples/Repeat.applescript -}
```

When you now type `80*-`, Typinator replaces this with a series of 80 dashes:

```
-----
```

Here is another example: Create an abbreviation `*paste` with the expansion:

```
{Samples/Repeat.applescript {clip}}
```

In this case, Typinator passes the contents of the clipboard as a parameter to the script, which then duplicates that as many times as you specify when you type the abbreviation. To see the script in effect, copy the text `"many, "` to the clipboard, and then type `12*paste`, and you will get:

```
many, many, many, many, many, many, many, many, many, many, many,
```

The *Multiply* script uses a similar trick. It multiplies the number entered before the abbreviation with the number given as the script's parameter. For example, try an abbreviation "mm with the following expansion:

```
{Samples/Multiply.applescript 25.4}
```

Then enter, for example, 4.5"mm to quickly convert 4.5 inch to mm (114.3).

Canceling expansions

When a script cannot create a meaningful result, you may want to tell Typinator that it should ignore the expansion altogether. To do this, you can use the special variable named `cancelExpansion`:

```
{{cancelExpansion=yes}}
```

The actual value does not matter. The presence of the variable is sufficient to tell Typinator that the expansion should not take place.

Canceling an expansion may be desirable for scripts that use the *Typinator_precedingText* environment variable to look at the text that was typed immediately before the current abbreviation. If this text does not contain any useful information, canceling the expansion may be a good idea. Typinator will then leave the typed abbreviation and the preceding text untouched.

Contributions are welcome

We make useful scripts available for download on our "[Download Extras](#)" web page. If you have developed a cool script that you wish to share with other Typinator users, please send it to [<typinator-support@ergonis.com>](mailto:typinator-support@ergonis.com).